Rules for integrands of the form $(e\,\mathrm{Trig}[a+b\,x])^m\,(f\,\mathrm{Trig}[c+d\,x])^n$

1. $\int \mathrm{Trig}[a+b\,x]\,\mathrm{Trig}[c+d\,x]\,dx$ when $b^2-d^2\neq 0$

   1: $\int \mathrm{Sin}[a+b\,x]\,\mathrm{Sin}[c+d\,x]\,dx$ when $b^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\mathrm{Sin}[v]\,\mathrm{Sin}[w] == \frac{1}{2}\,\mathrm{Cos}[v-w]-\frac{1}{2}\,\mathrm{Cos}[v+w]$

Rule: If $b^2-d^2\neq 0$, then

$$\int \mathrm{Sin}[a+b\,x]\,\mathrm{Sin}[c+d\,x]\,dx \longrightarrow \frac{\mathrm{Sin}[a-c+(b-d)\,x]}{2\,(b-d)}-\frac{\mathrm{Sin}[a+c+(b+d)\,x]}{2\,(b+d)}$$

Program code:

```
Int[sin[a_.+b_.*x_]*sin[c_.+d_.*x_],x_Symbol] :=
  Sin[a-c+(b-d)*x]/(2*(b-d)) - Sin[a+c+(b+d)*x]/(2*(b+d)) /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-d^2,0]
```

2: $\int \text{Cos}[a+b\,x]\,\text{Cos}[c+d\,x]\,dx$ when $b^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\text{Cos}[v]\,\text{Cos}[w]==\frac{1}{2}\text{Cos}[v-w]+\frac{1}{2}\text{Cos}[v+w]$

Rule: If $b^2-d^2\neq 0$, then

$$\int \text{Cos}[a+b\,x]\,\text{Cos}[c+d\,x]\,dx\longrightarrow \frac{\text{Sin}[a-c+(b-d)\,x]}{2\,(b-d)}+\frac{\text{Sin}[a+c+(b+d)\,x]}{2\,(b+d)}$$

Program code:

```
Int[cos[a_.+b_.*x_]*cos[c_.+d_.*x_],x_Symbol] :=
  Sin[a-c+(b-d)*x]/(2*(b-d)) + Sin[a+c+(b+d)*x]/(2*(b+d)) /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-d^2,0]
```

3: $\int \text{Sin}[a+b\,x]\,\text{Cos}[c+d\,x]\,dx$ when $b^2-d^2\neq 0$

Derivation: Algebraic expansion

Basis: $\text{Sin}[v]\,\text{Cos}[w]==\frac{1}{2}\text{Sin}[v+w]+\frac{1}{2}\text{Sin}[v-w]$

Rule: If $b^2-d^2\neq 0$, then

$$\int \text{Sin}[a+b\,x]\,\text{Cos}[c+d\,x]\,dx\longrightarrow -\frac{\text{Cos}[a-c+(b-d)\,x]}{2\,(b-d)}-\frac{\text{Cos}[a+c+(b+d)\,x]}{2\,(b+d)}$$

Program code:

```
Int[sin[a_.+b_.*x_]*cos[c_.+d_.*x_],x_Symbol] :=
  -Cos[a-c+(b-d)*x]/(2*(b-d)) - Cos[a+c+(b+d)*x]/(2*(b+d)) /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-d^2,0]
```

2. $\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2$

1. $\int (e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2$

1: $\int \operatorname{Cos}[a+b x]^2 (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land \left(\frac{p}{2} \in \mathbb{Z}^+ \lor p \notin \mathbb{Z}\right)$

Derivation: Algebraic expansion

Basis: $\operatorname{Cos}[z]^2 == \frac{1}{2} + \frac{1}{2} \operatorname{Cos}[2 z]$

Basis: $\operatorname{Sin}[z]^2 == \frac{1}{2} - \frac{1}{2} \operatorname{Cos}[2 z]$

Note: Although not necessary, this rule produces a slightly simpler antiderivative than the following rule.

Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land \left(\frac{p}{2} \in \mathbb{Z}^+ \lor p \notin \mathbb{Z}\right)$, then

$$\int \operatorname{Cos}[a+b x]^2 (g \operatorname{Sin}[c+d x])^p \, dx \rightarrow \frac{1}{2} \int (g \operatorname{Sin}[c+d x])^p \, dx + \frac{1}{2} \int \operatorname{Cos}[c+d x] (g \operatorname{Sin}[c+d x])^p \, dx$$

Program code:

```
Int[cos[a_.+b_.*x_]^2*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  1/2*Int[(g*Sin[c+d*x])^p,x] +
  1/2*Int[Cos[c+d*x]*(g*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && IGtQ[p/2,0]
```

```
Int[sin[a_.+b_.*x_]^2*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  1/2*Int[(g*Sin[c+d*x])^p,x] -
  1/2*Int[Cos[c+d*x]*(g*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && IGtQ[p/2,0]
```

2: $\int (e \operatorname{Cos}[a+b x])^m \operatorname{Sin}[c+d x]^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $\operatorname{Sin}[z] == 2 \operatorname{Cos}\left[\frac{z}{2}\right] \operatorname{Sin}\left[\frac{z}{2}\right]$

Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land p \in \mathbb{Z}$, then

$$\int (e \operatorname{Cos}[a+b x])^m \operatorname{Sin}[c+d x]^p \, dx \longrightarrow \frac{2^p}{e^p} \int (e \operatorname{Cos}[a+b x])^{m+p} \operatorname{Sin}[a+b x]^p \, dx$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_.*sin[c_.+d_.*x_]^p_.,x_Symbol] :=
  2^p/e^p*Int[(e*Cos[a+b*x])^(m+p)*Sin[a+b*x]^p,x] /;
FreeQ[{a,b,c,d,e,m},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && IntegerQ[p]
```

```
Int[(f_.*sin[a_.+b_.*x_])^n_.*sin[c_.+d_.*x_]^p_.,x_Symbol] :=
  2^p/f^p*Int[Cos[a+b*x]^p*(f*Sin[a+b*x])^(n+p),x] /;
FreeQ[{a,b,c,d,f,n},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && IntegerQ[p]
```

3. $\int (e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z}$

1: $\int (e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m + p - 1 == 0$

Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m + p - 1 == 0$, then

$$\int (e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^p \, dx \longrightarrow \frac{e^2 (e \operatorname{Cos}[a+b x])^{m-2} (g \operatorname{Sin}[c+d x])^{p+1}}{2 b g (p+1)}$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  e^2*(e*Cos[a+b*x])^(m-2)*(g*Sin[c+d*x])^(p+1)/(2*b*g*(p+1)) /;
FreeQ[{a,b,c,d,e,g,m,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && EqQ[m+p-1,0]
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -e^2*(e*Sin[a+b*x])^(m-2)*(g*Sin[c+d*x])^(p+1)/(2*b*g*(p+1)) /;
FreeQ[{a,b,c,d,e,g,m,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && EqQ[m+p-1,0]
```

2: $\int (e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m + 2 p + 2 == 0$

Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m + 2 p + 2 == 0$, then

$$\int (e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^p \, dx \longrightarrow -\frac{(e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^{p+1}}{b g m}$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -(e*Cos[a+b*x])^m*(g*Sin[c+d*x])^(p+1)/(b*g*m) /;
FreeQ[{a,b,c,d,e,g,m,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && EqQ[m+2*p+2,0]
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  (e*Sin[a+b*x])^m*(g*Sin[c+d*x])^(p+1)/(b*g*m) /;
FreeQ[{a,b,c,d,e,g,m,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && EqQ[m+2*p+2,0]
```

3. $\int (e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m > 1$

   1. $\int (e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m > 1 \land p < -1$

      1: $\int (e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m > 2 \land p < -1$

Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m > 2 \land p < -1$, then

$$\int (e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^p \, dx \longrightarrow$$
$$\frac{e^2 (e \operatorname{Cos}[a+b x])^{m-2} (g \operatorname{Sin}[c+d x])^{p+1}}{2 b g (p+1)} + \frac{e^4 (m+p-1)}{4 g^2 (p+1)} \int (e \operatorname{Cos}[a+b x])^{m-4} (g \operatorname{Sin}[c+d x])^{p+2} \, dx$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  e^2*(e*Cos[a+b*x])^(m-2)*(g*Sin[c+d*x])^(p+1)/(2*b*g*(p+1)) +
  e^4*(m+p-1)/(4*g^2*(p+1))*Int[(e*Cos[a+b*x])^(m-4)*(g*Sin[c+d*x])^(p+2),x] /;
FreeQ[{a,b,c,d,e,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,2] && LtQ[p,-1] && (GtQ[m,3] || EqQ[p,-3/2]) && IntegersQ[2
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -e^2*(e*Sin[a+b*x])^(m-2)*(g*Sin[c+d*x])^(p+1)/(2*b*g*(p+1)) +
  e^4*(m+p-1)/(4*g^2*(p+1))*Int[(e*Sin[a+b*x])^(m-4)*(g*Sin[c+d*x])^(p+2),x] /;
FreeQ[{a,b,c,d,e,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,2] && LtQ[p,-1] && (GtQ[m,3] || EqQ[p,-3/2]) && IntegersQ[2
```

      2: $\int (e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m > 1 \land p < -1 \land m + 2 p + 2 \neq 0$

Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m > 1 \land p < -1 \land m + 2 p + 2 \neq 0$, then

$$\int (e\,\text{Cos}[a+b\,x])^m\,(g\,\text{Sin}[c+d\,x])^p\,dx \longrightarrow$$
$$\frac{(e\,\text{Cos}[a+b\,x])^m\,(g\,\text{Sin}[c+d\,x])^{p+1}}{2\,b\,g\,(p+1)}+\frac{e^2\,(m+2\,p+2)}{4\,g^2\,(p+1)}\int (e\,\text{Cos}[a+b\,x])^{m-2}\,(g\,\text{Sin}[c+d\,x])^{p+2}\,dx$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  (e*Cos[a+b*x])^m*(g*Sin[c+d*x])^(p+1)/(2*b*g*(p+1)) +
  e^2*(m+2*p+2)/(4*g^2*(p+1))*Int[(e*Cos[a+b*x])^(m-2)*(g*Sin[c+d*x])^(p+2),x] /;
FreeQ[{a,b,c,d,e,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,1] && LtQ[p,-1] && NeQ[m+2*p+2,0] &&
  (LtQ[p,-2] || EqQ[m,2]) && IntegersQ[2*m,2*p]
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -(e*Sin[a+b*x])^m*(g*Sin[c+d*x])^(p+1)/(2*b*g*(p+1)) +
  e^2*(m+2*p+2)/(4*g^2*(p+1))*Int[(e*Sin[a+b*x])^(m-2)*(g*Sin[c+d*x])^(p+2),x] /;
FreeQ[{a,b,c,d,e,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,1] && LtQ[p,-1] && NeQ[m+2*p+2,0] &&
  (LtQ[p,-2] || EqQ[m,2]) && IntegersQ[2*m,2*p]
```

2: $\int (e\,\text{Cos}[a+b\,x])^m\,(g\,\text{Sin}[c+d\,x])^p\,dx$ when $b\,c-a\,d=0\,\land\,\frac{d}{b}=2\,\land\,p\notin\mathbb{Z}\,\land\,m>1\,\land\,m+2\,p\neq 0$

Rule: If $b\,c-a\,d=0\,\land\,\frac{d}{b}=2\,\land\,p\notin\mathbb{Z}\,\land\,m>1\,\land\,m+2\,p\neq 0$, then

$$\int (e\,\text{Cos}[a+b\,x])^m\,(g\,\text{Sin}[c+d\,x])^p\,dx \longrightarrow$$
$$\frac{e^2\,(e\,\text{Cos}[a+b\,x])^{m-2}\,(g\,\text{Sin}[c+d\,x])^{p+1}}{2\,b\,g\,(m+2\,p)}+\frac{e^2\,(m+p-1)}{m+2\,p}\int (e\,\text{Cos}[a+b\,x])^{m-2}\,(g\,\text{Sin}[c+d\,x])^{p}\,dx$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  e^2*(e*Cos[a+b*x])^(m-2)*(g*Sin[c+d*x])^(p+1)/(2*b*g*(m+2*p)) +
  e^2*(m+p-1)/(m+2*p)*Int[(e*Cos[a+b*x])^(m-2)*(g*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,g,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,1] && NeQ[m+2*p,0] && IntegersQ[2*m,2*p]
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -e^2*(e*Sin[a+b*x])^(m-2)*(g*Sin[c+d*x])^(p+1)/(2*b*g*(m+2*p)) +
  e^2*(m+p-1)/(m+2*p)*Int[(e*Sin[a+b*x])^(m-2)*(g*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,g,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,1] && NeQ[m+2*p,0] && IntegersQ[2*m,2*p]
```

4: $\int (e\,\mathrm{Cos}[a+b\,x])^m\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx$ when $b\,c-a\,d==0\,\wedge\,\frac{d}{b}==2\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m<-1\,\wedge\,m+2\,p+2\neq 0\,\wedge\,m+p+1\neq 0$

Rule: If $b\,c-a\,d==0\,\wedge\,\frac{d}{b}==2\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m<-1\,\wedge\,m+2\,p+2\neq 0\,\wedge\,m+p+1\neq 0$, then

$$\int (e\,\mathrm{Cos}[a+b\,x])^m\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx\,\longrightarrow$$
$$-\frac{(e\,\mathrm{Cos}[a+b\,x])^m\,(g\,\mathrm{Sin}[c+d\,x])^{p+1}}{2\,b\,g\,(m+p+1)}+\frac{m+2\,p+2}{e^2\,(m+p+1)}\int (e\,\mathrm{Cos}[a+b\,x])^{m+2}\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -(e*Cos[a+b*x])^m*(g*Sin[c+d*x])^(p+1)/(2*b*g*(m+p+1)) +
  (m+2*p+2)/(e^2*(m+p+1))*Int[(e*Cos[a+b*x])^(m+2)*(g*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,g,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && LtQ[m,-1] && NeQ[m+2*p+2,0] && NeQ[m+p+1,0] && IntegersQ[2*m,2*
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  (e*Sin[a+b*x])^m*(g*Sin[c+d*x])^(p+1)/(2*b*g*(m+p+1)) +
  (m+2*p+2)/(e^2*(m+p+1))*Int[(e*Sin[a+b*x])^(m+2)*(g*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,g,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && LtQ[m,-1] && NeQ[m+2*p+2,0] && NeQ[m+p+1,0] && IntegersQ[2*m,2*
```

5. $\int \mathrm{Cos}[a+b\,x]\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx$ when $b\,c-a\,d==0\,\wedge\,\frac{d}{b}==2\,\wedge\,p\notin\mathbb{Z}$

1: $\int \mathrm{Cos}[a+b\,x]\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx$ when $b\,c-a\,d==0\,\wedge\,\frac{d}{b}==2\,\wedge\,p\notin\mathbb{Z}\,\wedge\,p>0$

Rule: If $b\,c-a\,d==0\,\wedge\,\frac{d}{b}==2\,\wedge\,p\notin\mathbb{Z}\,\wedge\,p>0$, then

$$\int \operatorname{Cos}[a+b\,x]\,(g\operatorname{Sin}[c+d\,x])^p\,dx \to \frac{2\operatorname{Sin}[a+b\,x]\,(g\operatorname{Sin}[c+d\,x])^p}{d\,(2\,p+1)} + \frac{2\,p\,g}{2\,p+1}\int \operatorname{Sin}[a+b\,x]\,(g\operatorname{Sin}[c+d\,x])^{p-1}\,dx$$

– Program code:

```
Int[cos[a_.+b_.*x_]*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  2*Sin[a+b*x]*(g*Sin[c+d*x])^p/(d*(2*p+1)) + 2*p*g/(2*p+1)*Int[Sin[a+b*x]*(g*Sin[c+d*x])^(p-1),x] /;
FreeQ[{a,b,c,d,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[p,0] && IntegerQ[2*p]
```

```
Int[sin[a_.+b_.*x_]*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -2*Cos[a+b*x]*(g*Sin[c+d*x])^p/(d*(2*p+1)) + 2*p*g/(2*p+1)*Int[Cos[a+b*x]*(g*Sin[c+d*x])^(p-1),x] /;
FreeQ[{a,b,c,d,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[p,0] && IntegerQ[2*p]
```

2: $\int \operatorname{Cos}[a+b\,x]\,(g\operatorname{Sin}[c+d\,x])^p\,dx$ when $b\,c-a\,d=0\,\land\,\frac{d}{b}=2\,\land\,p\notin\mathbb{Z}\,\land\,p<-1$

– Rule: If $b\,c-a\,d=0\,\land\,\frac{d}{b}=2\,\land\,p\notin\mathbb{Z}\,\land\,p<-1$, then

$$\int \operatorname{Cos}[a+b\,x]\,(g\operatorname{Sin}[c+d\,x])^p\,dx \to \frac{\operatorname{Cos}[a+b\,x]\,(g\operatorname{Sin}[c+d\,x])^{p+1}}{2\,b\,g\,(p+1)} + \frac{2\,p+3}{2\,g\,(p+1)}\int \operatorname{Sin}[a+b\,x]\,(g\operatorname{Sin}[c+d\,x])^{p+1}\,dx$$

– Program code:

```
Int[cos[a_.+b_.*x_]*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  Cos[a+b*x]*(g*Sin[c+d*x])^(p+1)/(2*b*g*(p+1)) +
  (2*p+3)/(2*g*(p+1))*Int[Sin[a+b*x]*(g*Sin[c+d*x])^(p+1),x] /;
FreeQ[{a,b,c,d,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && LtQ[p,-1] && IntegerQ[2*p]
```

```
Int[sin[a_.+b_.*x_]*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -Sin[a+b*x]*(g*Sin[c+d*x])^(p+1)/(2*b*g*(p+1)) +
  (2*p+3)/(2*g*(p+1))*Int[Cos[a+b*x]*(g*Sin[c+d*x])^(p+1),x] /;
FreeQ[{a,b,c,d,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && LtQ[p,-1] && IntegerQ[2*p]
```

3: $\int \frac{\cos[a+b x]}{\sqrt{\sin[c+d x]}}\,dx$ when $b c - a d == 0 \land \frac{d}{b} == 2$

- Rule: If $b c - a d == 0 \land \frac{d}{b} == 2$, then

$$\int \frac{\cos[a+b x]}{\sqrt{\sin[c+d x]}}\,dx \longrightarrow -\frac{\text{ArcSin}[\cos[a+b x]-\sin[a+b x]]}{d}+\frac{\text{Log}\left[\cos[a+b x]+\sin[a+b x]+\sqrt{\sin[c+d x]}\right]}{d}$$

Program code:

```
Int[cos[a_.+b_.*x_]/Sqrt[sin[c_.+d_.*x_]],x_Symbol] :=
  -ArcSin[Cos[a+b*x]-Sin[a+b*x]]/d + Log[Cos[a+b*x]+Sin[a+b*x]+Sqrt[Sin[c+d*x]]]/d /;
FreeQ[{a,b,c,d},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2]
```

```
Int[sin[a_.+b_.*x_]/Sqrt[sin[c_.+d_.*x_]],x_Symbol] :=
  -ArcSin[Cos[a+b*x]-Sin[a+b*x]]/d - Log[Cos[a+b*x]+Sin[a+b*x]+Sqrt[Sin[c+d*x]]]/d /;
FreeQ[{a,b,c,d},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2]
```

6: $\int \frac{(g \sin[c+d x])^p}{\cos[a+b x]}\,dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z}$

- Derivation: Algebraic normalization

- Basis: $\frac{(g \sin[2 z])^p}{\cos[z]} == 2 g \sin[z] (g \sin[2 z])^{p-1}$

- Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z}$, then

$$\int \frac{(g \sin[c+d x])^p}{\cos[a+b x]}\,dx \longrightarrow 2 g \int \sin[a+b x] (g \sin[c+d x])^{p-1}\,dx$$

- Program code:

```
Int[(g_.*sin[c_.+d_.*x_])^p_/cos[a_.+b_.*x_],x_Symbol] :=
  2*g*Int[Sin[a+b*x]*(g*Sin[c+d*x])^(p-1),x] /;
FreeQ[{a,b,c,d,g,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && IntegerQ[2*p]
```

```
Int[(g_.*sin[c_.+d_.*x_])^p_/sin[a_.+b_.*x_],x_Symbol] :=
  2*g*Int[Cos[a+b*x]*(g*Sin[c+d*x])^(p-1),x] /;
FreeQ[{a,b,c,d,g,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && IntegerQ[2*p]
```

x: $\int (e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c-a d==0 \land \frac{d}{b}==2 \land p \notin \mathbb{Z} \land m+p \notin \mathbb{Z}$

Rule: If $b c-a d==0 \land \frac{d}{b}==2 \land p \notin \mathbb{Z} \land m+p \notin \mathbb{Z}$, then

$$\int (e \operatorname{Cos}[a+b x])^m (g \operatorname{Sin}[c+d x])^p \, dx \longrightarrow$$
$$-\frac{(e \operatorname{Cos}[a+b x])^{m+1} \operatorname{Sin}[a+b x] (g \operatorname{Sin}[c+d x])^p}{b e (m+p+1) \left(\operatorname{Sin}[a+b x]^2\right)^{\frac{p+1}{2}}} \operatorname{Hypergeometric2F1}\left[-\frac{p-1}{2}, \frac{m+p+1}{2}, \frac{m+p+3}{2}, \operatorname{Cos}[a+b x]^2\right]$$

–

Program code:

```
(* Int[(e_.*cos[a_.+b_.*x_])^m_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -(e*Cos[a+b*x])^(m+1)*Sin[a+b*x]*(g*Sin[c+d*x])^p/(b*e*(m+p+1)*(Sin[a+b*x]^2)^((p+1)/2))*
    Hypergeometric2F1[-(p-1)/2,(m+p+1)/2,(m+p+3)/2,Cos[a+b*x]^2] /;
FreeQ[{a,b,c,d,e,g,m,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && Not[IntegerQ[m+p]] *)
```

```
(* Int[(f_.*sin[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -Cos[a+b*x]*(f*Sin[a+b*x])^(n+1)*(g*Sin[c+d*x])^p/(b*f*(p+1)*(Sin[a+b*x]^2)^((n+p+1)/2))*
    Hypergeometric2F1[-(n+p-1)/2,(p+1)/2,(p+3)/2,Cos[a+b*x]^2] /;
FreeQ[{a,b,c,d,f,g,n,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && Not[IntegerQ[n+p]] *)
```

7: $\int (e \cos[a+b x])^m (g \sin[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b c - a d == 0 \land \frac{d}{b} == 2$, then $\partial_x \frac{(g \sin[c+d x])^p}{(e \cos[a+b x])^p \sin[a+b x]^p} == 0$

Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z}$, then

$$\int (e \cos[a+b x])^m (g \sin[c+d x])^p \, dx \to \frac{(g \sin[c+d x])^p}{(e \cos[a+b x])^p \sin[a+b x]^p} \int (e \cos[a+b x])^{m+p} \sin[a+b x]^p \, dx$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  (g*Sin[c+d*x])^p/((e*Cos[a+b*x])^p*Sin[a+b*x]^p)*Int[(e*Cos[a+b*x])^(m+p)*Sin[a+b*x]^p,x] /;
FreeQ[{a,b,c,d,e,g,m,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]]
```

```
Int[(f_.*sin[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  (g*Sin[c+d*x])^p/(Cos[a+b*x]^p*(f*Sin[a+b*x])^p)*Int[Cos[a+b*x]^p*(f*Sin[a+b*x])^(n+p),x] /;
FreeQ[{a,b,c,d,f,g,n,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]]
```

2. $\int (e\,\mathrm{Cos}[a+b\,x])^m\,(f\,\mathrm{Sin}[a+b\,x])^n\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx$ when $b\,c-a\,d==0\land\frac{d}{b}==2$

1: $\int \mathrm{Cos}[a+b\,x]^2\,\mathrm{Sin}[a+b\,x]^2\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx$ when $b\,c-a\,d==0\land\frac{d}{b}==2\land\left(\frac{p}{2}\in\mathbb{Z}^+\lor p\notin\mathbb{Z}\right)$

Derivation: Algebraic expansion

Basis: $\mathrm{Cos}[z]^2\,\mathrm{Sin}[z]^2==\frac{1}{4}-\frac{1}{4}\,\mathrm{Cos}[2\,z]^2$

Note: Although not necessary, this rule produces a slightly simpler antiderivative than the following rule.

Rule: If $b\,c-a\,d==0\land\frac{d}{b}==2\land\left(\frac{p}{2}\in\mathbb{Z}^+\lor p\notin\mathbb{Z}\right)$, then

$$\int \mathrm{Cos}[a+b\,x]^2\,\mathrm{Sin}[a+b\,x]^2\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx\longrightarrow\frac{1}{4}\int(g\,\mathrm{Sin}[c+d\,x])^p\,dx-\frac{1}{4}\int\mathrm{Cos}[c+d\,x]^2\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx$$

Program code:

```
Int[cos[a_.+b_.*x_]^2*sin[a_.+b_.*x_]^2*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  1/4*Int[(g*Sin[c+d*x])^p,x] -
  1/4*Int[Cos[c+d*x]^2*(g*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && IGtQ[p/2,0]
```

2: $\int (e\,\mathrm{Cos}[a+b\,x])^m\,(f\,\mathrm{Sin}[a+b\,x])^n\,\mathrm{Sin}[c+d\,x]^p\,dx$ when $b\,c-a\,d==0\,\land\,\frac{d}{b}==2\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: $\mathrm{Sin}[z]==2\,\mathrm{Cos}\left[\frac{z}{2}\right]\,\mathrm{Sin}\left[\frac{z}{2}\right]$

Rule: If $b\,c-a\,d==0\,\land\,\frac{d}{b}==2\,\land\,p\in\mathbb{Z}$, then

$$\int (e\,\mathrm{Cos}[a+b\,x])^m\,(f\,\mathrm{Sin}[a+b\,x])^n\,\mathrm{Sin}[c+d\,x]^p\,dx\,\longrightarrow\,\frac{2^p}{e^p\,f^p}\int (e\,\mathrm{Cos}[a+b\,x])^{m+p}\,(f\,\mathrm{Sin}[a+b\,x])^{n+p}\,dx$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_.*(f_.*sin[a_.+b_.*x_])^n_.*sin[c_.+d_.*x_]^p_.,x_Symbol] :=
  2^p/(e^p*f^p)*Int[(e*Cos[a+b*x])^(m+p)*(f*Sin[a+b*x])^(n+p),x] /;
FreeQ[{a,b,c,d,e,f,m,n},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && IntegerQ[p]
```

3. $\int (e\,\mathrm{Cos}[a+b\,x])^m\,(f\,\mathrm{Sin}[a+b\,x])^n\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx$ when $b\,c-a\,d==0\,\land\,\frac{d}{b}==2\,\land\,p\notin\mathbb{Z}$

1: $\int (e\,\mathrm{Cos}[a+b\,x])^m\,(f\,\mathrm{Sin}[a+b\,x])^n\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx$ when $b\,c-a\,d==0\,\land\,\frac{d}{b}==2\,\land\,p\notin\mathbb{Z}\,\land\,m+p-1==0$

Rule: If $b\,c-a\,d==0\,\land\,\frac{d}{b}==2\,\land\,p\notin\mathbb{Z}\,\land\,m+p-1==0$, then

$$\int (e\,\mathrm{Cos}[a+b\,x])^m\,(f\,\mathrm{Sin}[a+b\,x])^n\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx\,\longrightarrow\,\frac{e\,(e\,\mathrm{Cos}[a+b\,x])^{m-1}\,(f\,\mathrm{Sin}[a+b\,x])^{n+1}\,(g\,\mathrm{Sin}[c+d\,x])^p}{b\,f\,(n+p+1)}$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_.*(f_.*sin[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  e*(e*Cos[a+b*x])^(m-1)*(f*Sin[a+b*x])^(n+1)*(g*Sin[c+d*x])^p/(b*f*(n+p+1)) /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && EqQ[m+p+1,0]
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_*(f_.*cos[a_.+b_.*x_])^n_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -e*(e*Sin[a+b*x])^(m-1)*(f*Cos[a+b*x])^(n+1)*(g*Sin[c+d*x])^p/(b*f*(n+p+1)) /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && EqQ[m+p+1,0]
```

2: $\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m+n+2p+2 == 0 \land m+p+1 \neq 0$

Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m+n+2p+2 == 0 \land m+p+1 \neq 0$, then

$$\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx \longrightarrow -\frac{(e \operatorname{Cos}[a+b x])^{m+1} (f \operatorname{Sin}[a+b x])^{n+1} (g \operatorname{Sin}[c+d x])^p}{b e f (m+p+1)}$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_.*(f_.*sin[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -(e*Cos[a+b*x])^(m+1)*(f*Sin[a+b*x])^(n+1)*(g*Sin[c+d*x])^p/(b*e*f*(m+p+1)) /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && EqQ[m+n+2*p+2,0] && NeQ[m+p+1,0]
```

3. $\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m > 1$

1. $\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m > 1 \land p < -1$

1: $\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m > 3 \land p < -1 \land n + p + 1 \neq 0$

Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m > 3 \land p < -1 \land n + p + 1 \neq 0$, then

$$\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx \longrightarrow$$

$$\frac{e^2 (e \operatorname{Cos}[a+b x])^{m-2} (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^{p+1}}{2 b g (n+p+1)} + \frac{e^4 (m+p-1)}{4 g^2 (n+p+1)} \int (e \operatorname{Cos}[a+b x])^{m-4} (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^{p+2} \, dx$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_*(f_.*sin[a_.+b_.*x_])^n_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  e^2*(e*Cos[a+b*x])^(m-2)*(f*Sin[a+b*x])^n*(g*Sin[c+d*x])^(p+1)/(2*b*g*(n+p+1)) +
  e^4*(m+p-1)/(4*g^2*(n+p+1))*Int[(e*Cos[a+b*x])^(m-4)*(f*Sin[a+b*x])^n*(g*Sin[c+d*x])^(p+2),x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,3] && LtQ[p,-1] && NeQ[n+p+1,0] && IntegersQ[2*m,2*n,2*p
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_*(f_.*cos[a_.+b_.*x_])^n_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -e^2*(e*Sin[a+b*x])^(m-2)*(f*Cos[a+b*x])^n*(g*Sin[c+d*x])^(p+1)/(2*b*g*(n+p+1)) +
  e^4*(m+p-1)/(4*g^2*(n+p+1))*Int[(e*Sin[a+b*x])^(m-4)*(f*Cos[a+b*x])^n*(g*Sin[c+d*x])^(p+2),x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,3] && LtQ[p,-1] && NeQ[n+p+1,0] && IntegersQ[2*m,2*n,2*p
```

2: $\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \wedge \frac{d}{b} == 2 \wedge p \notin \mathbb{Z} \wedge m > 1 \wedge p < -1 \wedge m+n+2p+2 \neq 0 \wedge n+p+1 \neq 0$

- Rule: If $b c - a d == 0 \wedge \frac{d}{b} == 2 \wedge p \notin \mathbb{Z} \wedge m > 1 \wedge p < -1 \wedge m+n+2p+2 \neq 0 \wedge n+p+1 \neq 0$, then

$$\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx \longrightarrow$$
$$\frac{(e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^{p+1}}{2 b g (n+p+1)} + \frac{e^2 (m+n+2p+2)}{4 g^2 (n+p+1)} \int (e \operatorname{Cos}[a+b x])^{m-2} (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^{p+2} \, dx$$

- Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_*(f_.*sin[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  (e*Cos[a+b*x])^m*(f*Sin[a+b*x])^n*(g*Sin[c+d*x])^(p+1)/(2*b*g*(n+p+1)) +
  e^2*(m+n+2*p+2)/(4*g^2*(n+p+1))*Int[(e*Cos[a+b*x])^(m-2)*(f*Sin[a+b*x])^n*(g*Sin[c+d*x])^(p+2),x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,1] && LtQ[p,-1] && NeQ[m+n+2*p+2,0] && NeQ[n+p+1,0] &&
  IntegersQ[2*m,2*n,2*p] && (LtQ[p,-2] || EqQ[m,2] || EqQ[m,3])
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_*(f_.*cos[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -(e*Sin[a+b*x])^m*(f*Cos[a+b*x])^n*(g*Sin[c+d*x])^(p+1)/(2*b*g*(n+p+1)) +
  e^2*(m+n+2*p+2)/(4*g^2*(n+p+1))*Int[(e*Sin[a+b*x])^(m-2)*(f*Cos[a+b*x])^n*(g*Sin[c+d*x])^(p+2),x] /;
FreeQ[{a,b,c,d,e,f,g,n},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,1] && LtQ[p,-1] && NeQ[m+n+2*p+2,0] && NeQ[n+p+1,0] &&
  IntegersQ[2*m,2*n,2*p] && (LtQ[p,-2] || EqQ[m,2] || EqQ[m,3])
```

2: $\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m > 1 \land n < -1 \land n + p + 1 \neq 0$

Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m > 1 \land n < -1 \land n + p + 1 \neq 0$, then

$$\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx \longrightarrow$$

$$\frac{e (e \operatorname{Cos}[a+b x])^{m-1} (f \operatorname{Sin}[a+b x])^{n+1} (g \operatorname{Sin}[c+d x])^p}{b f (n+p+1)} + \frac{e^2 (m+p-1)}{f^2 (n+p+1)} \int (e \operatorname{Cos}[a+b x])^{m-2} (f \operatorname{Sin}[a+b x])^{n+2} (g \operatorname{Sin}[c+d x])^p \, dx$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_*(f_.*sin[a_.+b_.*x_])^n_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  e*(e*Cos[a+b*x])^(m-1)*(f*Sin[a+b*x])^(n+1)*(g*Sin[c+d*x])^p/(b*f*(n+p+1)) +
  e^2*(m+p-1)/(f^2*(n+p+1))*Int[(e*Cos[a+b*x])^(m-2)*(f*Sin[a+b*x])^(n+2)*(g*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,1] && LtQ[n,-1] && NeQ[n+p+1,0] && IntegersQ[2*m,2*n,2*p
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_*(f_.*cos[a_.+b_.*x_])^n_*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -e*(e*Sin[a+b*x])^(m-1)*(f*Cos[a+b*x])^(n+1)*(g*Sin[c+d*x])^p/(b*f*(n+p+1)) +
  e^2*(m+p-1)/(f^2*(n+p+1))*Int[(e*Sin[a+b*x])^(m-2)*(f*Cos[a+b*x])^(n+2)*(g*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,1] && LtQ[n,-1] && NeQ[n+p+1,0] && IntegersQ[2*m,2*n,2*p
```

3: $\int (e\,\mathrm{Cos}[a+b\,x])^m\,(f\,\mathrm{Sin}[a+b\,x])^n\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx$ when $b\,c-a\,d=0\,\land\,\frac{d}{b}=2\,\land\,p\notin\mathbb{Z}\,\land\,m>1\,\land\,m+n+2\,p\neq 0$

–

Rule: If $b\,c-a\,d=0\,\land\,\frac{d}{b}=2\,\land\,p\notin\mathbb{Z}\,\land\,m>1\,\land\,m+n+2\,p\neq 0$, then

$$\int (e\,\mathrm{Cos}[a+b\,x])^m\,(f\,\mathrm{Sin}[a+b\,x])^n\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx\,\longrightarrow$$
$$\frac{e\,(e\,\mathrm{Cos}[a+b\,x])^{m-1}\,(f\,\mathrm{Sin}[a+b\,x])^{n+1}\,(g\,\mathrm{Sin}[c+d\,x])^p}{b\,f\,(m+n+2\,p)}+\frac{e^2\,(m+p-1)}{m+n+2\,p}\int (e\,\mathrm{Cos}[a+b\,x])^{m-2}\,(f\,\mathrm{Sin}[a+b\,x])^n\,(g\,\mathrm{Sin}[c+d\,x])^p\,dx$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_*(f_.*sin[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  e*(e*Cos[a+b*x])^(m-1)*(f*Sin[a+b*x])^(n+1)*(g*Sin[c+d*x])^p/(b*f*(m+n+2*p)) +
  e^2*(m+p-1)/(m+n+2*p)*Int[(e*Cos[a+b*x])^(m-2)*(f*Sin[a+b*x])^n*(g*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,1] && NeQ[m+n+2*p,0] && IntegersQ[2*m,2*n,2*p]
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_*(f_.*cos[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -e*(e*Sin[a+b*x])^(m-1)*(f*Cos[a+b*x])^(n+1)*(g*Sin[c+d*x])^p/(b*f*(m+n+2*p)) +
  e^2*(m+p-1)/(m+n+2*p)*Int[(e*Sin[a+b*x])^(m-2)*(f*Cos[a+b*x])^n*(g*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && GtQ[m,1] && NeQ[m+n+2*p,0] && IntegersQ[2*m,2*n,2*p]
```

4. $\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m < -1$

1: $\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m < -1 \land n > 0 \land p > 0 \land m + n + 2 p \neq 0$

Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m < -1 \land n > 0 \land p > 0 \land m + n + 2 p \neq 0$, then

$$\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx \longrightarrow$$

$$-\frac{f (e \operatorname{Cos}[a+b x])^{m+1} (f \operatorname{Sin}[a+b x])^{n-1} (g \operatorname{Sin}[c+d x])^p}{b e (m+n+2 p)} + \frac{2 f g (n+p-1)}{e (m+n+2 p)} \int (e \operatorname{Cos}[a+b x])^{m+1} (f \operatorname{Sin}[a+b x])^{n-1} (g \operatorname{Sin}[c+d x])^{p-1} \, dx$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_*(f_.*sin[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -f*(e*Cos[a+b*x])^(m+1)*(f*Sin[a+b*x])^(n-1)*(g*Sin[c+d*x])^p/(b*e*(m+n+2*p)) +
  2*f*g*(n+p-1)/(e*(m+n+2*p))*Int[(e*Cos[a+b*x])^(m+1)*(f*Sin[a+b*x])^(n-1)*(g*Sin[c+d*x])^(p-1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && LtQ[m,-1] && GtQ[n,0] && GtQ[p,0] && NeQ[m+n+2*p,0] &&
  IntegersQ[2*m,2*n,2*p]
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_*(f_.*cos[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  f*(e*Sin[a+b*x])^(m+1)*(f*Cos[a+b*x])^(n-1)*(g*Sin[c+d*x])^p/(b*e*(m+n+2*p)) +
  2*f*g*(n+p-1)/(e*(m+n+2*p))*Int[(e*Sin[a+b*x])^(m+1)*(f*Cos[a+b*x])^(n-1)*(g*Sin[c+d*x])^(p-1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && LtQ[m,-1] && GtQ[n,0] && GtQ[p,0] && NeQ[m+n+2*p,0] &&
  IntegersQ[2*m,2*n,2*p]
```

2:

$\int (e \cos[a+b x])^m (f \sin[a+b x])^n (g \sin[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m < -1 \land n > 0 \land p < -1 \land m+n+2p+2 \neq 0 \land m+p+1 \neq 0$

- Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m < -1 \land n > 0 \land p < -1 \land m+n+2p+2 \neq 0 \land m+p+1 \neq 0$, then

$$\int (e \cos[a+b x])^m (f \sin[a+b x])^n (g \sin[c+d x])^p \, dx \longrightarrow$$

$$-\frac{(e \cos[a+b x])^{m+1} (f \sin[a+b x])^{n+1} (g \sin[c+d x])^p}{b e f (m+p+1)} + \frac{f (m+n+2p+2)}{2 e g (m+p+1)} \int (e \cos[a+b x])^{m+1} (f \sin[a+b x])^{n-1} (g \sin[c+d x])^{p+1} \, dx$$

- Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_*(f_.*sin[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -(e*Cos[a+b*x])^(m+1)*(f*Sin[a+b*x])^(n+1)*(g*Sin[c+d*x])^p/(b*e*f*(m+p+1)) +
  f*(m+n+2*p+2)/(2*e*g*(m+p+1))*Int[(e*Cos[a+b*x])^(m+1)*(f*Sin[a+b*x])^(n-1)*(g*Sin[c+d*x])^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && LtQ[m,-1] && GtQ[n,0] && LtQ[p,-1] && NeQ[m+n+2*p+2,0] &&
  NeQ[m+p+1,0] && IntegersQ[2*m,2*n,2*p]
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_*(f_.*cos[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  (e*Sin[a+b*x])^(m+1)*(f*Cos[a+b*x])^(n+1)*(g*Sin[c+d*x])^p/(b*e*f*(m+p+1)) +
  f*(m+n+2*p+2)/(2*e*g*(m+p+1))*Int[(e*Sin[a+b*x])^(m+1)*(f*Cos[a+b*x])^(n-1)*(g*Sin[c+d*x])^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && LtQ[m,-1] && GtQ[n,0] && LtQ[p,-1] && NeQ[m+n+2*p+2,0] &&
  NeQ[m+p+1,0] && IntegersQ[2*m,2*n,2*p]
```

3: $\int (e\,\text{Cos}[a+b\,x])^m\,(f\,\text{Sin}[a+b\,x])^n\,(g\,\text{Sin}[c+d\,x])^p\,dx$ when $b\,c-a\,d==0\,\wedge\,\frac{d}{b}==2\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m<-1\,\wedge\,m+n+2\,p+2\neq 0\,\wedge\,m+p+1\neq 0$

Rule: If $b\,c-a\,d==0\,\wedge\,\frac{d}{b}==2\,\wedge\,p\notin\mathbb{Z}\,\wedge\,m<-1\,\wedge\,m+n+2\,p+2\neq 0\,\wedge\,m+p+1\neq 0$, then

$$\int (e\,\text{Cos}[a+b\,x])^m\,(f\,\text{Sin}[a+b\,x])^n\,(g\,\text{Sin}[c+d\,x])^p\,dx\,\longrightarrow$$
$$-\frac{(e\,\text{Cos}[a+b\,x])^{m+1}\,(f\,\text{Sin}[a+b\,x])^{n+1}\,(g\,\text{Sin}[c+d\,x])^p}{b\,e\,f\,(m+p+1)}+\frac{m+n+2\,p+2}{e^2\,(m+p+1)}\int (e\,\text{Cos}[a+b\,x])^{m+2}\,(f\,\text{Sin}[a+b\,x])^n\,(g\,\text{Sin}[c+d\,x])^p\,dx$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_*(f_.*sin[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -(e*Cos[a+b*x])^(m+1)*(f*Sin[a+b*x])^(n+1)*(g*Sin[c+d*x])^p/(b*e*f*(m+p+1)) +
  (m+n+2*p+2)/(e^2*(m+p+1))*Int[(e*Cos[a+b*x])^(m+2)*(f*Sin[a+b*x])^n*(g*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && LtQ[m,-1] && NeQ[m+n+2*p+2,0] && NeQ[m+p+1,0] &&
  IntegersQ[2*m,2*n,2*p]
```

```
Int[(e_.*sin[a_.+b_.*x_])^m_*(f_.*cos[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  (e*Sin[a+b*x])^(m+1)*(f*Cos[a+b*x])^(n+1)*(g*Sin[c+d*x])^p/(b*e*f*(m+p+1)) +
  (m+n+2*p+2)/(e^2*(m+p+1))*Int[(e*Sin[a+b*x])^(m+2)*(f*Cos[a+b*x])^n*(g*Sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && LtQ[m,-1] && NeQ[m+n+2*p+2,0] && NeQ[m+p+1,0] &&
  IntegersQ[2*m,2*n,2*p]
```

x: $\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx$ when $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m + p \notin \mathbb{Z} \land n + p \notin \mathbb{Z}$

Rule: If $b c - a d == 0 \land \frac{d}{b} == 2 \land p \notin \mathbb{Z} \land m + p \notin \mathbb{Z} \land n + p \notin \mathbb{Z}$, then

$$\int (e \operatorname{Cos}[a+b x])^m (f \operatorname{Sin}[a+b x])^n (g \operatorname{Sin}[c+d x])^p \, dx \longrightarrow$$

$$-\frac{(e \operatorname{Cos}[a+b x])^{m+1} (f \operatorname{Sin}[a+b x])^{n+1} (g \operatorname{Sin}[c+d x])^p}{b e f (m+p+1) \left(\operatorname{Sin}[a+b x]^2\right)^{\frac{n+p+1}{2}}} \operatorname{Hypergeometric2F1}\left[-\frac{n+p-1}{2}, \frac{m+p+1}{2}, \frac{m+p+3}{2}, \operatorname{Cos}[a+b x]^2\right]$$

Program code:

```
(* Int[(e_.*cos[a_.+b_.*x_])^m_*(f_.*sin[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  -(e*Cos[a+b*x])^(m+1)*(f*Sin[a+b*x])^(n+1)*(g*Sin[c+d*x])^p/(b*e*f*(m+p+1)*(Sin[a+b*x]^2)^((n+p+1)/2))*
    Hypergeometric2F1[-(n+p-1)/2,(m+p+1)/2,(m+p+3)/2,Cos[a+b*x]^2] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]] && Not[IntegerQ[m+p]] && Not[IntegerQ[n+p]] *)
```

5: $\int (e\,Cos[a+b\,x])^m\,(f\,Sin[a+b\,x])^n\,(g\,Sin[c+d\,x])^p\,dx$ when $b\,c-a\,d==0\,\wedge\,\frac{d}{b}==2\,\wedge\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b\,c-a\,d==0\,\wedge\,\frac{d}{b}==2$, then $\partial_x\frac{(g\,Sin[c+d\,x])^p}{(e\,Cos[a+b\,x])^p\,(f\,Sin[a+b\,x])^p}==0$

Rule: If $b\,c-a\,d==0\,\wedge\,\frac{d}{b}==2\,\wedge\,p\notin\mathbb{Z}$, then

$$\int (e\,Cos[a+b\,x])^m\,(f\,Sin[a+b\,x])^n\,(g\,Sin[c+d\,x])^p\,dx\,\longrightarrow\,\frac{(g\,Sin[c+d\,x])^p}{(e\,Cos[a+b\,x])^p\,(f\,Sin[a+b\,x])^p}\int (e\,Cos[a+b\,x])^{m+p}\,(f\,Sin[a+b\,x])^{n+p}\,dx$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_.*(f_.*sin[a_.+b_.*x_])^n_.*(g_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  (g*Sin[c+d*x])^p/((e*Cos[a+b*x])^p*(f*Sin[a+b*x])^p)*Int[(e*Cos[a+b*x])^(m+p)*(f*Sin[a+b*x])^(n+p),x] /;
FreeQ[{a,b,c,d,e,f,g,m,n,p},x] && EqQ[b*c-a*d,0] && EqQ[d/b,2] && Not[IntegerQ[p]]
```

3: $\int (e\,Cos[a+b\,x])^m\,Sin[c+d\,x]\,dx$ when $b\,c-a\,d==0\,\wedge\,\frac{d}{b}==Abs[m+2]$

Rule: If $b\,c-a\,d==0\,\wedge\,\frac{d}{b}==Abs[m+2]$, then

$$\int (e\,Cos[a+b\,x])^m\,Sin[c+d\,x]\,dx\,\longrightarrow\,-\frac{(m+2)\,(e\,Cos[a+b\,x])^{m+1}\,Cos[(m+1)\,(a+b\,x)]}{d\,e\,(m+1)}$$

Program code:

```
Int[(e_.*cos[a_.+b_.*x_])^m_.*sin[c_.+d_.*x_],x_Symbol] :=
  -(m+2)*(e*Cos[a+b*x])^(m+1)*Cos[(m+1)*(a+b*x)]/(d*e*(m+1)) /;
FreeQ[{a,b,c,d,e,m},x] && EqQ[b*c-a*d,0] && EqQ[d/b,Abs[m+2]]
```